
*=============Class_D_Current_Sources============
One common application for a Class D output stage is
to drive LEDs. But keeping output current constant does
require some feedback, and Class D output stages provide
some stability challenges when used in feedback.

This feedback example is limited to an open loop gain of 100 
in the feedback loop. A high feedback loop makes it possible
to keep the voltage across RL constant. To check this, a 1volt 
peak to peak 50Hz signal is applied to VCM. 

=======================================================================
CLASS_D_CurrentSource
*
*              _________________________________/\  /\  /\_______
*       ____  |                    |       |      \/  \/ RT =1   |
*     _|_   | | |\          IREF   |   ^   |                     |
*    /   \  | |_|+\               _|_ /|\  |  CCOMP          VD3 |
*   /B_TRI\ |   |  \_____        / _ \ |  _|_                    |
*   \     / |   |  /     |       \/ \/    ___                    |
*    \___/  |___|-/ BVG  |       /\_/\     |     ____|\  BOTA    |
*     _|_       |/       |       \___/    _|_   |    |-\  _  _   |
*     ///                |        _|_     ///   |    |  \/ \/ \__|
*      __________________|        \ /           |    |  /\_/\_/
*     |                            V VCC        |   _|+/
*     |  VCC _/\  /\  /\_ VP                    |  | |/
*     |    _|_  \/  \/   |                      |  |
*     |   /VCC\ RP     <-                  ___  |  |     ___
*     |   \___/      ||___ MP1            |OUT|_|  |____|IS |
*     |    _|_    ___||                   |___|         |___|
*     |    ///    |  ||_           _  _  _  |    RS=30m   |
*     |           |     |  ____   / \/ \/ \ |             |   R1 = 5
*     |___________|    _|_|VOUT|__| () () |_|__/\  /\  /\_|_/\  /\  /\__
*           VG    |__||   |____|     L1      |   \/  \/       \/  \/   _|_
*                    ||__                    |                        /_  \ VCM
*                    ||-> MN1  Freq = 9 K    |    C1                 // \  \
*              RN       |      Z=5.8Ohms     |__||__ 3u              \   \//
*           _/\  /\  /\_|                       || _|_                \___/
*         _|_  \/  \/  VN                          ///                 _|_
*         ///                                                          ///
*
.OPTIONS  GMIN=1p       METHOD=trap   ABSTOL=1u     TEMP=27   srcsteps = 1  gminsteps = 1 
.OPTIONS  RELTOL=.001   ABSTOL=1n     VNTOL=1u      ITL4=500  ITL1=400   
*=========Create_Signal==================
VT        VT     0      DC     0      PWL(   0      0    1      1)
Vfreq     Vfreq  0      DC     100k
B_TRI     TRI    0      V =    acos( cos(6.283185*V(VFreq)*V(VT)) )/3.141592
IREF      VCC    VD     .15u
CCOMP     VD     0      30p    IC=.5
VB        VB     0      DC     .5
RB        VD     VBS    96000k
RBS       VBS    VB     1
RT        VD3    VD     1
B_OTA     VD3     0     I      =1u*tanh((tanh(V(OUT)-V(IS))*10)*1)
B_VD2     VD2    0      V      = tanh(u(v(VD))*v(VD) )     
BVG       VG     0      V =    5*u( v(TRI) -v(VD2)) 
VCC       VCC    0      DC     5
RP        VCC    VP     1u
RN        VN     0      1u
MN1       VOUT   VG     VN     0      NMOSC   W=90000u    L=1u 
MP1       VOUT   VG     VP     VCC    PMOSC   W=90000u    L=1u 
L1        VOUT   OUT    100u
C1        OUT    0      3u
RS        OUT    IS     30m
Rout      IS     VCM    5
VCM       VCM    0      DC     0      SIN(   .5    .5   50      )



*TRAN     TSTEP  TSTOP  TSTART TMAX   ?UIC?
.tran      .1u    20m     1m     .1u   UIC
*=========The_CMOS_Model_Files==================================
.model    NMOSC              NMOS(Level=  1  Cbs=2f Cbd=2f)      
.model    PMOSC              PMOS(Level=  1  Cbs=2f Cbd=2f)
.model    DD                 D(     IS=3.15e-18 )
.control
run
set       pensize  = 2
plot      out -is  ylimit 0 20m

=======================================================================
A 30mOhm value is being used for RS, and the intention 
is to maintain a 500mA current through it.  

=======================================================================

let       iota   = mean(vd-vd3)
let       irb    = mean(vbs-vb)
let       ivdt   = iota + irb
let       vdave  = mean(vd)
let       vd2ave = mean(vd2)
plot      vcm      out 
echo      "iota =$&iota irb =$&irb ivdt =$&ivdt vdave =$&vdave vd2ave =$&vd2ave    "
.endc
.end

=======================================================================
LEDs need to be more current source driven. It is nice
to be able to do this at a high power efficiency.  

=======================================================================



The alternative is to use a transistor collector or drain,
which will waste a lot of heat.

===========Full_Netlist_For_Copy_Paste=======================
CLASS_D_CurrentSource_AC
*
*              _________________________________/\  /\  /\_______
*       ____  |                    |       |      \/  \/ RT =1   |
*     _|_   | | |\          IREF   |   ^   |                     |
*    /   \  | |_|+\               _|_ /|\  |  CCOMP          VD3 |
*   /B_TRI\ |   |  \_____        / _ \ |  _|_                    |
*   \     / |   |  /     |       \/ \/    ___                    |
*    \___/  |___|-/ BVG  |       /\_/\     |     ____|\  BOTA    |
*     _|_       |/       |       \___/    _|_   |    |-\  _  _   |
*     ///                |        _|_     ///   |    |  \/ \/ \__|
*      __________________|        \ /           |    |  /\_/\_/
*     |                            V VCC        |   _|+/
*     |  VCC _/\  /\  /\_ VP                    |  | |/
*     |    _|_  \/  \/   |                      |  |
*     |   /VCC\ RP     <-                  ___  |  |     ___
*     |   \___/      ||___ MP1            |OUT|_|  |____|IS |
*     |    _|_    ___||                   |___|         |___|
*     |    ///    |  ||_           _  _  _  |    RS=10m   |
*     |           |     |  ____   / \/ \/ \ |             |   R1 = 5
*     |___________|    _|_|VOUT|__| () () |_|__/\  /\  /\_|_/\  /\  /\__
*           VG    |__||   |____|     L1      |   \/  \/       \/  \/   _|_
*                    ||__                    |                        /_  \ VCM
*                    ||-> MN1  Freq = 9 K    |    C1                 // \  \
*              RN       |      Z=5.8Ohms     |__||__ 3u              \   \//
*           _/\  /\  /\_|                       || _|_                \___/
*         _|_  \/  \/  VN                          ///                 _|_
*         ///                                                          ///
*
.OPTIONS  GMIN=1p       METHOD=trap   ABSTOL=1u     TEMP=27   srcsteps = 1  gminsteps = 1 
.OPTIONS  RELTOL=.001   ABSTOL=1n     VNTOL=1u      ITL4=500  ITL1=400   
*=========Create_Signal==================
VT        VT     0      DC     0      PWL(   0      0    1      1)
Vfreq     Vfreq  0      DC     100k
B_TRI     TRI    0      V =    acos( cos(6.283185*V(VFreq)*V(VT)) )/3.141592
IREF      VCC    VD     .15u
CCOMP     VD     0      30p    IC=.5
VB        VB     0      DC     .5
RB        VD     VBS    96000k
RBS       VBS    VB     1
RT        VD3    VD     1
B_OTA     VD3     0     I      =1u*tanh((tanh(V(OUT)-V(IS))*10)*1)



B_VD2     VD2    0      V      = tanh(u(v(VD))*v(VD) )     
BVG       VG     0      V =    5*u( v(TRI) -v(VD2)) 
VCC       VCC    0      DC     5
RP        VCC    VP     1u
RN        VN     0      1u
MN1       VOUT   VG     VN     0      NMOSC   W=90000u    L=1u 
MP1       VOUT   VG     VP     VCC    PMOSC   W=90000u    L=1u 
L1        VOUT   OUT    100u
C1        OUT    0      3u
RS        OUT    IS     30m
Rout      IS     VCM    5
VCM       VCM    0      DC     0      SIN(   .5    .5   50      )

*TRAN     TSTEP  TSTOP  TSTART TMAX   ?UIC?
.tran     .1u    20m     1m     .1u   UIC

*=========The_CMOS_Model_Files==================================
.model    NMOSC              NMOS(Level=  1  Cbs=2f Cbd=2f)      
.model    PMOSC              PMOS(Level=  1  Cbs=2f Cbd=2f)
.model    DD                 D(     IS=3.15e-18 )
 
.control
run
set       pensize = 2
plot      out -is  ylimit 0 20m

let       iota = mean(vd-vd3)
let       irb =   mean(vbs-vb)
let       ivdt = iota + irb
let       vdave =mean(vd)
let       vd2ave =mean(vd2)
plot      vcm  out
  
echo      "iota =$&iota irb =$&irb ivdt =$&ivdt vdave =$&vdave vd2ave =$&vd2ave    "

.endc

.end

9.16.10_1.30PM
dsauersanjose@aol.com
Don Sauer 


